# Maintainer: Maxime Gauduin <alucryd@archlinux.org>

pkgname=python-flask-marshmallow
pkgver=0.15.0
pkgrel=1
pkgdesc='Flask + marshmallow for beautiful APIs'
url=https://flask-marshmallow.readthedocs.io/
arch=(any)
license=(MIT)
depends=(
  python-flask
  python-marshmallow
)
makedepends=(
  git
  python-setuptools
)
_tag=b36e14ae413efdf54cbb1b99f62bc572c0ec04e3
source=(git+https://github.com/marshmallow-code/flask-marshmallow.git#tag=${_tag})
b2sums=(SKIP)

pkgver() {
  cd flask-marshmallow
  git describe --tags
}

build() {
  cd flask-marshmallow
  python setup.py build
}

package() {
  cd flask-marshmallow
  python setup.py install --root="${pkgdir}" --optimize=1 --skip-build
  install -Dm 644 LICENSE -t "${pkgdir}"/usr/share/licenses/python-flask-marshmallow
}

# vim: ts=2 sw=2 et:
